# Maintainer: Filipe Laíns (FFY00) <lains@archlinux.org>

pkgname=python-colorlog
pkgver=6.7.0
pkgrel=2
pkgdesc='A colored formatter for the python logging module'
arch=('any')
url='https://github.com/borntyping/python-colorlog'
license=('MIT')
depends=('python')
makedepends=('python-build' 'python-installer' 'python-setuptools' 'python-wheel')
source=("$pkgname-$pkgver.tar.gz::$url/archive/v$pkgver.tar.gz")
sha512sums=('cd1a82dbb0a02106078804eac81b1d1a20ab7ed6c60909057e9d904c98531c88c1ed0c57ed0ad5d8a53ff5f3fe5fc695a9ce9783cbc073e699a59d330af54a2c')

build() {
  cd $pkgname-$pkgver

  python -m build -nw
}

package() {
  cd $pkgname-$pkgver

  python -m installer -d "$pkgdir" dist/*.whl

  install -Dm 644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}

# vim:set ts=2 sw=2 et:
